The Read-Bajraktarevic Operator

A preprint version of a “Mathematical graphics” column from Mathematica in Education and Research.

Mark McClure

mcmcclur@unca.edu

Department of Mathematics
University of North Carolina at Asheville
Asheville, NC 28804

Abstract

The Read-Bajraktarevic operator is a fairly abstract construct from the area of functional equations. It has been used in recent years in connection with self-affine functions. This yield a nice example of mathematical graphics illustrating abstract mathematics.

■ Mathematica Initializations

1. A functional equation

A major objective of this column is the exposition and illustration of high-level mathematics using computer graphics. Frequently, the graphics themselves are a lure whose aesthetic appeal, hopefully, draws the interest and curiosity of some readers. Explanation of algorithms then proceeds after interest is piqued. In this issue's column, we turn this formula around by starting with a very abstract statement; the question then becomes, “what could that possibly mean?” In this case, we can use computer graphics to illustrate the abstract statement.

The theorem of interest comes from the subject of functional equations. This theorem was proved in the 1950s by two mathematicians working independently. In recent years, it has been used to describe functions with fractal properties, for example in [1]. The main tool arising from this work has been dubbed the Read-Bajraktarevic operator in honor of the original discoverers.

Consider the following theorem:
Let I be a closed interval and suppose that $b : I \overset{\text{onto}}{\longrightarrow} I$ and $v : I \times \mathbb{R} \to \mathbb{R}$ are continuous. Furthermore, suppose there is an $r \in (0, 1)$ such that for every $x \in I$ and $y_1, y_2 \in \mathbb{R}$, $|v(x, y_1) - v(x, y_2)| \leq r |y_1 - y_2|$. Define an operator $\Phi : C_\infty(I) \to C_\infty(I)$ by $\Phi f(x) = v(x, f(b(x)))$. Then Φ is a contractive operator on C_∞ and, therefore, has a unique fixed point $f_\Phi \in C_\infty(I)$.

Let's make sure we understand the notation. First, note that $C_\infty(I)$ simply refers to the set of all bounded, continuous, and real valued functions defined on the interval I. The distance between any two such functions f and g is simply $\max\{|f(x) - g(x)| : x \in I\}$. Given continuous functions $b : I \overset{\text{onto}}{\longrightarrow} I$ and $v : I \times \mathbb{R} \to \mathbb{R}$, the theorem describes how to generate a function $\Phi : C_\infty(I) \to C_\infty(I)$. The theorem furthermore states that this function is contractive.

We could now try to illustrate this by simply jumping in with *Mathematica*. The standard way to find the fixed point of a contraction is via iteration. Thus, we choose functions b and v which satisfy the hypotheses for some interval I, we set up the corresponding operator Φ and iterate from an arbitrary starting function. Suppose for example that $b(x) = x^2$ and $v(x, y) = 3(x+y)/4$. These functions certainly satisfy the hypotheses for the unit interval $I = [0, 1]$. Here's how to set up the operator Φ and iterate it 8 times starting from the zero function $f_0(x) = 0$. We also **`Expand`** the results and place them in **`TableForm`** to reveal the convergence more clearly.

```
In[1]:=  b[x_] := x^2;
         v[x_, y_] := 3 (x + y) / 4;
         Φ[f_] := v[x, f /. x → b[x]];
         NestList[Φ, 0, 8] // Expand // TableForm
```

```
Out[4]//TableForm=
```

0

$\frac{3x}{4}$

$\frac{3x}{4} + \frac{9x^2}{16}$

$\frac{3x}{4} + \frac{9x^2}{16} + \frac{27x^4}{64}$

$\frac{3x}{4} + \frac{9x^2}{16} + \frac{27x^4}{64} + \frac{81x^8}{256}$

$\frac{3x}{4} + \frac{9x^2}{16} + \frac{27x^4}{64} + \frac{81x^8}{256} + \frac{243x^{16}}{1024}$

$\frac{3x}{4} + \frac{9x^2}{16} + \frac{27x^4}{64} + \frac{81x^8}{256} + \frac{243x^{16}}{1024} + \frac{729x^{32}}{4096}$

$\frac{3x}{4} + \frac{9x^2}{16} + \frac{27x^4}{64} + \frac{81x^8}{256} + \frac{243x^{16}}{1024} + \frac{729x^{32}}{4096} + \frac{2187x^{64}}{16384}$

$\frac{3x}{4} + \frac{9x^2}{16} + \frac{27x^4}{64} + \frac{81x^8}{256} + \frac{243x^{16}}{1024} + \frac{729x^{32}}{4096} + \frac{2187x^{64}}{16384} + \frac{6561x^{128}}{65536}$

From these computations, it's fairly clear that the fixed point f_Φ may be represented as the following power series:

$$\sum_{n=0}^{\infty} \left(\frac{3}{4}\right)^n x^{2^n}.$$

While illustrative of the basic idea, the previous example is not particularly exciting. It turns out that the Read-Bajraktarevic operator may be used to generate graphs with fractal properties. For this to happen, the function b needs to mix up the domain a bit. For example, suppose we choose b to be the logistic function $b(x) = 4x(1-x)$. This is a standard example in chaos theory of a function which "mixes up" the unit interval. In fact, as it's graph shows, it maps both the first and second half of the unit interval onto the entire unit interval. Under iteration, the unit interval is folded over onto itself multiple times.

```
In[5]:=  b[x_] := 4 x (1 - x);
         Plot[b[b[x]], {x, 0, 1}];
```



We now define $v(x, y)$ as before, iterate the corresponding operator Φ say 10 times and plot the result.

```
In[7]:= v[x_, y_] := 3 (x + y) / 4;
        Φ[f_] := v[x, f /. x → b[x]];
        app = Nest[Φ, 0, 10];
        Plot[app, {x, 0, 1},
         PlotRange → All, PlotPoints → 100];
```



Note that convergence is not as easy to see analytically, in this example. While the iterates are polynomials, there is not an obvious limiting series.

```
In[11]:=  NestList[Φ, 0, 4] // Expand // TableForm
```

Out[11]//TableForm=

$$0$$
$$\frac{3x}{4}$$
$$3x - \frac{9x^2}{4}$$
$$\frac{39x}{4} - 36x^2 + 54x^3 - 27x^4$$
$$30x - \frac{1845x^2}{4} + 3456x^3 - 13392x^4 + 28512x^5 - 33696x^6 + 20736x^7 - 5184x^8$$

In retrospect, this should not be too surprising. If the limit were a power series, it would necessarily be everywhere differentiable; the graph, however, indicates that this is not the case here. The convergence can certainly be illustrated graphically, however.

```
In[12]:= b[x_] := 4 x (1 - x);
         v[x_, y_] := 3 (x + y) / 4;
         Φ[f_] := v[x, f /. x → b[x]];
         apps = NestList[Φ, 0, 9];
         pics = Plot[#, {x, 0, 1},
             PlotRange → {0, 2},
             PlotPoints → 100,
             FrameTicks → {{0, 1}, {0, 1, 2}, {}, {}},
             DisplayFunction → Identity] & /@ apps;
         Show[GraphicsArray[Partition[Drop[pics, 3], 2]]];
```



2. Self-affine graphs

While we have illustrated the convergence described in the Read-Bajraktarevic theorem in a couple of examples, it is still not clear what type of functions we might expect to generate. In fact, the theorem has a very concrete, geometrical interpretation. The function b is a transformation of the domain of any $f \in C_\infty(I)$, while the function v returns values in the range of f. Expressed geometrically, if G_f is the graph of the function f over the interval I in the plane, then b transforms G_f in the x-direction, while v transforms G_f in the y-direction. The graph of a fixed point of Φ therefore displays some type invariance under these geometrical transformations.

Consider, for example, the function $f(x) = x^2$. Then $f(2\ x)/4 = (2\ x)^2/4 = 4\ x^2/4 = x^2 = f(x)$. We can express this in terms of the Read-Bajraktarevic operator as follows.

```
In[18]:= b[x_] := 2 x;
         v[x_, y_] := y / 4;
         Φ[f_] := v[x, f /. x → b[x]];
         Φ[x^2]

Out[21]= x^2
```

Geometrically, this states that the graph of $f(x) = x^2$ is invariant under the affine transformation $T(x, y) = (x/2, y/4)$. Note that the fixed point is not unique in this example.

```
In[22]:= Φ[0]

Out[22]= 0
```

There is no contradiction here since the hypotheses of the theorem are not quite satisfied. In particular, there is no interval which is invariant under b. Nonetheless, the function $f(x) = x^2$ is an example of an important class of functions called self-affine functions and the hypotheses of the Read-Bajraktarevic theorem may be weakened slightly to include this class.

To explain this, we must define the self-affine functions. An affine transformation of the plane is simply a function $T : \mathbb{R} \to \mathbb{R}$ of the form

$$T\begin{pmatrix} x \\ y \end{pmatrix} = A\begin{pmatrix} x \\ y \end{pmatrix} + b$$

where A is a two dimensional matrix and b is a two dimensional vector. Now suppose that f is a real valued function defined on a closed interval I and denote its graph by $G_f = \{(x, f(x)) : x \in I\}$. We say that f is a self-affine function if G_f is a self-affine set. This means that G_f is composed of smaller affine images of itself. More precisely, there are affine transformations $T_1, T_2, \ldots, T_m$ such that

$$G_f = \bigcup_{i=1}^{m} T_i(G_f).$$

The list of transformations $T_1, T_2, \ldots, T_m$ is usually called an iterated function system.

Consider for example the graph of $f(x) = x^2$ over the unit interval $I = [0, 1]$. This is a self-affine function with respect to the iterated function system

$$T_1\begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} 1/2 & 0 \\ 0 & 1/4 \end{pmatrix}\begin{pmatrix} x \\ y \end{pmatrix} \text{ and } T_2\begin{pmatrix} x \\ y \end{pmatrix} = \begin{pmatrix} 1/2 & 0 \\ 1/2 & 1/4 \end{pmatrix}\begin{pmatrix} x \\ y \end{pmatrix} + \begin{pmatrix} 0 \\ 1/4 \end{pmatrix}.$$

As described above, T_1 maps G_f to the portion of G_f over $[0, 1/2]$. The second transformation T_2 maps G_f to the portion of G_f over $[1/2, 1]$. To see the second part, suppose that (x, x^2) is any point on G_f and note that

$$T_2\begin{pmatrix} x \\ x^2 \end{pmatrix} = \begin{pmatrix} 1/2 & 0 \\ 1/2 & 1/4 \end{pmatrix}\begin{pmatrix} x \\ x^2 \end{pmatrix} + \begin{pmatrix} 1/2 \\ 1/4 \end{pmatrix} = \begin{pmatrix} \frac{1}{2}x + \frac{1}{2} \\ (\frac{1}{2}x + \frac{1}{2})^2 \end{pmatrix}.$$

Now our question is how to fit this type of function into the Read-Bajraktarevic framework. In order to do so, we need the following somewhat more general version of their theorem:

Let I be a closed interval and suppose that $b : I \xrightarrow{\text{onto}} I$ and $v : I \times \mathbb{R} \to \mathbb{R}$. Furthermore, suppose there is an $r \in (0, 1)$ such that for every $x \in I$ and $y_1, y_2 \in \mathbb{R}$, $|v(x, y_1) - v(x, y_2)| \leq r\,|y_1 - y_2|$. Define an operator $\Phi : C_\infty(I) \to C_\infty(I)$ by

$\Phi f(x) = v(x, f(b(x)))$ and suppose there is a subset of C_∞ which is invariant under Φ. Then Φ has a unique fixed point $f_\Phi \in C_\infty(I)$.

Note that b and v are no longer assumed to be continuous but we explicitly assume that there is a subset of C_∞ which is invariant under Φ, ie. the subset maps into itself under the action of Φ. Of course, if b and v are continuous, then all of C_∞ is invariant under Φ. Thus the hypotheses of the original theorem have been weakened.

Now that b and v are no longer required to be continuous, we may define them in a piecewise manner. In the following code, b maps each half of the unit interval onto the whole unit interval and v is defined differently for values of x in these two domains. In this way, the Read-Bajraktarevic operator can mimic the behavior of an iterated function system.

```
In[23]:= b[x_] := Mod[2 x, 1];
         v[x_, y_] := { y / 4              0 ≤ x < 1 / 2
                        y / 4 + x - 1 / 4  1 / 2 ≤ x < 1 ;
         Φ[f_] := v[x, f /. x → b[x]];
         fΦ = Φ[x^2]
```

$$\text{Out[26]=} \begin{cases} \frac{1}{4}\,\text{Mod}[2\,x,\,1]^2 & 0 \le x < \frac{1}{2} \\ -\frac{1}{4} + x + \frac{1}{4}\,\text{Mod}[2\,x,\,1]^2 & \frac{1}{2} \le x < 1 \end{cases}$$

Note that any function continuous function f which satisfies $\Phi f(1/2) = \lim_{x\to 1/2^-} \Phi f(x)$ after application of Φ will again be continuous. In particular, $f(x) = x^2$ satisfies this property. In fact, f_Φ is really just x^2 in disguise.

```
In[27]:= Plot[fΦ, {x, 0, 1}];
```



For our final example, we look at a more typical self-affine function which consists of three pieces.

```
In[28]:= b[x_] := Mod[3 x, 1];
         v[x_, y_] := ⎡    2 y / 3        0 ≤ x < 1 / 3
                      ⎨ -y / 3 + 2 / 3  1 / 3 ≤ x < 2 / 3;
                      ⎣ 2 y / 3 + 1 / 3   2 / 3 ≤ x < 1
         Φ[f_] := v[x, f /. x → b[x]];
         fΦ = Nest[Φ, x, 7];
         Plot[fΦ, {x, 0, 1}];
```



References

[1] Massopust, Peter R. *Fractal functions, fractal surfaces, and wavelets*. Academic Press, Inc., San Diego, CA, 1994.